【问题描述】

几十年前全世界就流行一种数字游戏，至今仍有人乐此不疲．在中国我们把这种游戏称为“算24点”。您作为游戏者将得到4个1~9之间的自然数作为操作数，而您的任务是对这4个操作数进行适当的算术运算，要求运算结果等于24。

您可以使用的运算只有：+，-，*，/，您还可以使用（）来改变运算顺序。注意：所有的中间结果须是整数，所以一些除法运算是不允许的（例如，(2*2)/4是合法的，2*(2/4)是不合法的）。下面我们给出一个游戏的具体例子：

若给出的4个操作数是：1、2、3、7，则一种可能的解答是1+2+3*7=24。
【输入】

只有一行，四个1到9之间的自然数。
【输出】

如果有解的话，只要输出一个解，输出的是三行数据，分别表示运算的步骤。其中第一行是输入的两个数和一个运算符和运算后的结果，第二行是第一行的结果和一个输入的数据、运算符、运算后的结果；第三行是第二行的结果和输入的一个数、运算符和“=24”。如果两个操作数有大小的话则先输出大的。

如果没有解则输出“No answer!”
【样例】

point24.in

point24.out

2+1=3






7*3=21






21+3=24

【算法分析】

计算24点主要应用四种运算．开始状态有四个操作数，一个运算符对应两个操作数，所以一开始选择两个操作数分别对四个操作符进行循环检测，每一次运算后产生了新的数，两个数运算变成一个数，整体是少了一个操作数，所以四个数最终是三次运算。由于操作的层数比较少(只有三层)，所以可以用回溯的算法来进行检测，当找到一个解时便结束查找。如果所有的情况都找过后仍然没有，则输出无解的信息。
 program E1_3; {point24}
type arr=array [1..4] of integer;
var i,result,n,len:integer;
    d:arr;
    r:array [1..3,1..4] of integer;
    infile,outfile:text;
procedure print;
var i,j:integer;
begin
  assign(outfile,'point24.out');
rewrite(outfile);
  for i:=1 to 3 do
  begin
      for j:=1 to 3 do
        if j<>2 then write(outfile,r[i,j])
          else case r[i,j] of
                  1:write(outfile,'+');
                  2:write(outfile,'-');
                  3:write(outfile,'*');
                  4:write(outfile,'/')
                end;
       writeln(outfile,'=',r[i,4])
    end;
  close(outfile);
end;
procedure try(k:integer;d:arr);
var a,b,i,j,l,t:integer;
    e:arr;
begin
  if k=1 then if d[1]=24 then begin print;halt end else
     else
     begin
       for i:=1 to k-1 do
        for j:=i+1 to k do
         begin
           a:=d[i]; b:=d[j];
           if a<b then begin t:=a;a:=b;b:=t end;
           t:=0;
           for l:=1 to k do if (l<>i) and (l<>j) then begin t:=t+1;e[t]:=d[l] end;
           r[5-k,1]:=a;
           r[5-k,3]:=b;
           r[5-k,4]:=-1;
           for l:=1 to 4 do
            begin
             case l of
               1:r[5-k,4]:=a+b;
               2:r[5-k,4]:=a-b;
               3:r[5-k,4]:=a*b;
               4:if b<>0  then if a mod b=0 then r[5-k,4]:=a div b
             end;
            r[5-k,2]:=l;
            if r[5-k,4]<>-1 then
                 begin
                  e[t+1]:=r[5-k,4];
                  try(k-1,e)
                 end
           end
        end
     end;
end;
begin
  assign(infile,'point24.in');
  reset(infile);
  for i:=1 to 4 do read(infile,d[i]);
  close(infile);
  try(4,d);
  assign(outfile,'point24.out');
  rewrite(outfile);
  writeln(outfile,'No answer!');
  close(outfile);
end.

